Capes et Xcas

Renee.Degraeve@wanadoo.fr Bernard.Parisse@ujf-grenoble.fr

2010

Table des matières

On donne dans ce texte quelques informations pour les candidats au Capes désirant s'informer sur l'utilisation de `Xcas` pendant les oraux.

1 Comparaison des fonctionnalités des logiciels.

La liste des logiciels est disponible sur le site du jury de capes de maths :
`http://capes-math.org/`
Il s'agit de logiciels libres ou gratuits, sauf les émulateurs de calculatrices (il existe des versions d'essai d'un mois). Pour les émulateurs de calculatrices, on note l'absence d'émulateur TI89/92/Voyage 200 (cet oubli sera peut-être corrigé ?). Merci de signaler des erreurs éventuelles.

Logiciel	Géo 2-d	3-d	Graphes	Tableur	Suites	CAS[1]	Algo
Algobox							+
ClassPad Manager	++		2d 3d	++	+	++	+=
Geogebra	+++		2d	+	=	+	
Geoplan/Geospace	+=	+	2d 3d		=		
Maxima			2d 3d		=	+++	++
OpenOffice.org			2d	+++	=		
Python							+++
Scilab			2d 3d	(stats)			++
Sinequanon	+		2d	(stats)	+		
TI-NSpire CAS TE	++		2d	++	+	++	++
TI-SmartView 83+	Cabri Jr		2d	CellSheet	+		
Xcas	++	+	2d 3d	++	+=	+++	+++

Bien entendu il y a une part de subjectivité dans le nombre de `+`. Précisons un peu :

– en géométrie 2-d : Geogebra est clairement le leader en termes de facilité d'utilisation, rendu et nombre d'utilisateurs. Geoplan bénéficie encore d'une implantation historique, mais voit très probablement son nombre d'utilisateurs diminuer au profit de Geogebra. `Xcas` est le seul de la liste qui permet de faire des figures avec des coordonnées exactes et donc des démonstrations en géométrie analytique.

– en géométrie 3-d : l'interface de Geospace (en particulier la gestion du point de vue 3-d et des surfaces) me parait très peu intuitive (c'est peut-être subjectif).

– pour le tableur et les stats au tableur, l'interface d'OpenOffice est évidemment largement en avance sur les concurrents. Par contre, les tableurs de `Xcas`, TI et Casio sont des tableurs formels exacts, et le tableur de `Xcas` et de Geogebra peuvent interagir avec la géométrie.

– pour les suites récurrentes, la représentation en est simple avec TI, Casio, `Xcas`. Le tableau de valeur peut être obtenu avec n'importe quel tableur. La résolution formelle d'une relation de récurrence est possible dans des cas simples avec `Xcas` et Maxima.

– pour le calcul formel, ce sont `Xcas` et Maxima qui ont le plus de fonctionnalités, `Xcas` étant plus rapide, mais TI et Casio sont largement suffisant au niveau lycée et STS.

– en algorithmique : Algobox parait vraiment limité à l'apprentissage du B.A.-BA, pas utilisable en 1ère S ou au-delà, mais dispose de la possibilité d'exécuter en pas-à-pas. Le langage des Classpad est un peu moins restreint, mais n'a pas de fonction non algébrique, le langage de TI est assez complet, mais il n'y a pas d'outils de mise au point. Le langage de scilab ou maxima est complet, mais le premier est numérique uniquement. Python et `Xcas` proposent un langage complet avec débogueur interactif, Python est un langage généraliste, alors que `Xcas` est orienté maths ce qui selon la situation peut être un avantage ou un inconvénient.

Maitriser un minimum Geogebra ou Open Office calc est certainement à la portée de tout le monde. `Xcas` nécessite un peu plus d'apprentissage, mais avec un seul logiciel, on est sur de ne faire aucune impasse (et faire l'impasse sur l'algorithmique est probablement risqué !). C'est vrai dans une moindre mesure des deux autres intégrés (émulateurs TI et Casio), mais avec l'inconvénient d'utiliser un logiciel propriétaire (avec seulement une version d'essai gratuite).

2 Suggestions oral I

On donne ici quelques noms de commande et des pistes pour les candidats souhaitant illustrer avec l'outil informatique une lecon de l'oral 1. Ces pistes ne sont naturellement pas exhaustives et ne doivent évidemment pas prendre le pas sur la lecon elle-même, cela peut par exemple être une commande de tracé de graphe de fonction, une ou deux commandes pertinentes pour illustrer par un exemple et ne doit pas prendre beaucoup de temps de préparation (en tout cas pas plus d'1/2 heure sur les 2h1/2)

1. Techniques de dénombrement, Coefficients binomiaux :
   `comb, perm, factorial`
2. Expérience aléatoire, probabilité, probabilité conditionnelle :
   `alea`
3. Loi binomiale :
   `binomial, binomial_cdf, binomial_icdf`
4. Loi de Poisson :
   `poisson, poisson_cdf, poisson_icdf`,
5. Loi normale :
   `normald, normald_cdf, normald_icdf`
6. Statistiques descriptives à 1 variable :
   `mean, stddev, variance`,
   `median, quartile1, quartile3, moustache`,
   voir aussi le menu Cmds▶Proba-stats de `Xcas` et Math▶stats de son tableur
7. Série statistiques à 2 variables :
   `correlation, covariance`,
   `linear_regression, linear_regression_plot` et variantes,
   voir aussi le menu Cmds▶Proba-stats de `Xcas` et Maths▶stats de son tableur
8. Estimation ponctuelle ou par intervalle de confiance d'un paramètre, test d'hypothèses :
   `binomial_icdf, normald_icdf, poisson_icdf`.
   Cf. dans l'aide de `Xcas` le manuel de statistiques :
   Aide▶Manuels▶Tableur,statistiques.

9. Multiples, diviseurs, division euclidienne :
   `iquo, irem, idivis`
   Cf. dans l'aide de `Xcas` le manuel de programmation et un exemple :
   Aide▶Manuels▶Programmation, et Aide▶Exemples▶arit▶diviseur
10. PGCD, PPCM de 2 entiers naturels :
    `gcd, lcm`,
    voir aussi dans l'aide de `Xcas` l'algorithme d'Euclide :
    Aide▶Manuels▶Programmation qu'on peut copier et exécuter en mode pas à pas.
11. Egalité de Bézout :
    `iegcd`,
    voir aussi dans l'aide le manuel de programmation de `Xcas` et l'exemple dans :
    Aide▶Exemples▶arit▶bezout
12. Nombres premiers, décomposition d'un entier en produit de facteurs premiers :
    `isprime, nextprime, prevprime, ifactor`
    Voir aussi des algorithmes simples dans le manuel de programmation de `Xcas`.
13. Congruences dans Z :
    `irem, smod, powmod, ichinrem`
14. Equations du second degré à coeffs réels ou complexes / Fonctions polynômes du second degré :
    `forme_canonique`, `solve, csolve`,
    `factor, cfactor`
15. Module et argument d'un nombre complexe :
    `abs, arg, exp2trig, trig2exp`,
    géométrie en entrant les complexes sous forme exponentielle
16. Transformations planes et nombres complexes : `affixe, similitude`, les fonctions de géométrie 2-d de `Xcas` travaillent avec des nombres complexes
17. Exemples d'utilisation des nombres complexes.
    Démonstration du théorème de Napoléon. Point de concours des hauteurs (sujet du 29 juin 2006).
18. Algèbre linéaire en STS :
    menu CAS▶Algèbre linéaire et Cmds▶Alglin
19. Systèmes d'équations et d'inéquations :
    `solve, linsolve`
20. Droites du plan :
    `droite, parallele, perpendiculaire`,
    `inter_unique`, `equation, parameq`
21. Droites et plans de l'espace :
    `droite, plan`,
    `parallele, perpendiculaire`,
    `equation, parameq`, `inter_unique`
22. Droites remarquables du triangle :
    menu Geo▶Lignes

23. Le cercle :
menu Geo▶Cercles, `tangent`, `equation, parameq`
24. Solides de l'espace :
menu Geo▶Solides et Geo▶Solides_Platon
25. Barycentre :
`barycentre`, `isobarycentre`
26. Produit scalaire :
`produit_scalaire`
27. Trigonométrie :
menu Expression
28. Produit vectoriel, produit mixte :
`cross, dot`
29. Homothétie, translation, isométries planes, similitudes planes :
Menu Geo▶Transformations
30. Problèmes de constructions géométriques, problèmes de lieux :
manuels Aide▶Manuels▶Geométrie et Aide▶Manuels▶Exercices
31. Convergence de suites réelles :
(`limit`)
32. Suites arithmétiques, suites géométriques
33. Suites de terme général $a^n, n^p, \ln(n), a \in \mathbb{R}^{+*}, p \in \mathbb{N}, n \in \mathbb{N}^*$
34. Suites de nombres réels définis par une récurrence :
menu du tableur Maths▶Suites▶Suite récurrente,
(`plotseq`, `tableseq`), `rsolve`
35. Problèmes conduisant à l'étude de suites :
Aide▶Exemples▶analyse▶newton.xws
36. Limite d'une fonction réelle de variable réelle :
`limit`
37. Fonctions logarithme
38. Fonctions exponentielles
39. Croissance comparée $e^x, x^a, \ln(x)$:
`plot, limit`
40. Courbes planes paramétriques :
`plotparam, parameq`
41. Intégrales, primitives :
`int`
42. Techniques de calcul d'intégrales :
`ibpdv`, `ibpu`, `subst`,
`tlin, trig2exp, lin`,`halftan, partfrac`,
voir aussi le manuel d'exercices : Aide▶Manuels▶Exercices
43. Équations différentielles :
`desolve`, `plotode, plotfield`

44. Problèmes conduisant à la résolution d'équations différentielles
45. Problèmes conduisant à l'étude de fonctions
46. Développements limités :
    `series, divpc`
47. Séries numériques :
    `sum`, manuel de programmation
48. Séries de Fourier :
    `fourier_an, fourier_bn, fourier_cn`
49. Transformée de Laplace :
    `laplace, ilaplace`
50. Courbes de Bézier
51. Exemples d'études de courbe :
    manuel exercices
52. Aires :
    `area`, `plotarea`
53. Exemples d'algorithmes :
    manuel de programmation
54. Exemples d'utilisation d'un tableur : manuel du tableur
55. Application des mathématiques à d'autres disciplines :
    par exemple une des sessions de Aide►Exemples►climat

3 Oral 2

Attention, il n'y a que 2h1/2 de préparation pour la partie mathématique **et** la partie agir en fonctionnaires, il ne faut donc pas passer beaucoup de temps de préparation sur une illustration informatique (mais bien évidemment si une illustration est explicitement demandée, par exemple une figure, il faut quand même se donner le temps de la faire).

Sessions Xcas correspondant aux exercices de l'oral 2006
Des exemples de sessions correspondant aux énoncés donnés à l'oral 2 du Capes 2006 sont disponibles depuis le menu Aide, Exemples, Capes2006.

Corrigés d'exercices épreuve expérimentale TS.
Ces exercices sont dans l'esprit de certains exercices qui peuvent être proposés pour l'oral 2. Voir la page
`http://www-fourier.ujf-grenoble.fr/~parisse/irem.html#bacs`

Corrections d'exercices donnés à l'oral 2 du capes 2008
La suite de cette section donne des solutions avec `Xcas` de ces exercices mais n'est en aucun cas un corrigé-type de ce qu'on attend d'un candidat le jour de l'oral 2 : cela aurait pu constituer une partie de l'exposé du candidat. Le classement par thèmes peut également servir de pistes d'illustrations pour l'oral 1.

3.1 Thème : Équations diffŕentielles

– L'exercice proposé au candidat
  On considère les deux équations différentielles définies sur $]-\frac{\pi}{2},\frac{\pi}{2}[$ par :

$$(E) \quad y' + (1+\tan(x))y = \cos(x) \qquad\qquad (E0) \quad y' + y = 1$$

  1. Donner l'ensemble des solutions de $(E0)$.
  2. Soient g une fonction dérivable sur $]-\frac{\pi}{2},\frac{\pi}{2}[$.
     On pose $f(x) = g(x)\cos(x)$. Démontrer que la fonction f est solution de (E) si et seulement si la fonction g est solution de $(E0)$.
  3. Déterminer la solution f de (E) telle que $f(0) = 0$.

– Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Indiquer, pour chacune des questions de l'exercice, les savoirs mis en jeu.
  Q.2) Présenter une solution de la question 2) de l'exercice telle que vous la donneriez à des élèves de Terminale.

– Une présentation et une solution avec `Xcas`
  1. Tout d'abord il faut faire comprendre à des élèves de Terminale ce qu'est une équation différentielle et pour cela leur faire observer le champ des tangentes des deux équations proposées.
     On commence par l'équation $(E_0) \quad y' = 1-y$ et on tape :

```
fieldplot(1-y,[x,y])
```

     On remarque qu'en chaque point (x,y), la pente de la tangente vaut $1-y$.
     Puis on tape :

```
interactive_plotode(1-y,[x,y])
```

     pour voir les solutions se dessiner en suivant les directions indiquées par les tangentes.
     On obtient :



Attention
Le premier argument de `fieldplot` et de `interactive_plotode` est la fonction `h(x,y)` qui est la valeur de `y'`.
On continue par l'équation $(E) \quad y' = \cos(x) - (1 + \tan(x))y$. Pour des facilités de mise en œuvre, on ouvre un écran de géométrie et on utilise le menu :
`Graphe->Slopefield/Ode(2d)`
qui ouvre une boite de dialogues que l'on remplit :
- on coche `Field` car on veut le tracé du champ des tangentes,
- on coche `||=1` car on veut que ces tangentes soient normalisées,
- on met la valeur de y' : `cos(x)-(1+tan(x))*y`,
- le nom des variables `x` et `y`,
- on met les différentes valeurs du cadrage et on met un pas.

Puis on tape sur `OK`.
On obtient le champ des tangentes de (E) et la commande :

```
fieldplot(cos(x)-(1+tan(x))*y,[x,y],normalize)
```

Puis on clique en un point dans l'écran de géométrie et on obtient une courbe intégrale de (E) passant par ce point et la commande :

```
A :=plotode(cos(x)-(1+tan(x))*y,[x,y])
```

etc...
On obtient :



Attention
Lorsque le champ des tangentes est normalisé, les tangentes sont centrées au point de contact, et si le champ des tangentes n'est pas normalisé, les tangentes ont pour origine le point de contact et indique la vitesse en chaque point.

2. On fait résoudre l'équation $(E0)$ par `Xcas`, on tape :

   ```
   desolve(y'+y-1,x,y)
   ```

   On obtient

   ```
   (exp(x)+c_0)/(exp(x))
   ```

3. On fait résoudre l'équation (E) par `Xcas`, on tape :

   ```
   desolve(y'+(1+tan(x))*y-cos(x),x,y)
   ```

   On obtient

   ```
   (cos(x)*exp(x)+c_0*cos(x))/(exp(x))
   ```

   On remarque qu'on obtient les solutions de (E) en multipliant les solutions de $(E0)$ par $\cos(x)$.

4. On cherche la solution f de (E) telle que $f(0) = 0$, avec `Xcas`, on tape :

   ```
   desolve([y'+(1+tan(x))*y-cos(x),y(0)=0],x,y)
   ```

   On obtient :

   ```
   [(cos(x)*exp(x)-cos(x))/(exp(x))]
   ```

5. On peut faire faire par `Xcas` le changement de fonction inconnue dans $(E0)$ ($g(x) = f(x)/\cos(x)$).
   On tape :

   ```
   E2 :=factor(expand(subst(g(x)'+g(x)-1,g(x)=f(x)/cos(x))))
   ```

   On obtient l'équation vérifiée par $f(x)$:

   ```
   (cos(x)*diff(f(x),x)+cos(x)*f(x)-cos(x)^2+
   sin(x)*f(x))/ (cos(x)^2)
   ```

   Il faut arranger cette expression pour remplacer la fonction $\sin$ en les fonctions $\tan$ et $\cos$ puis factoriser : On tape :

   ```
   normal(sin2costan(E2))
   ```

   On obtient l'équation vérifiée par $f(x)$:

   ```
   (-cos(x)+diff(f(x),x)+f(x)*tan(x)+f(x))/(cos(x))
   ```

   Ou bien, on transforme (E).
   On remplace $f(x)\ \cos(x) * g(x))$ et on transforme la fonction $\tan$ en les fonctions $\sin$ et $\cos$ et on tape :

   ```
   factor(tan2sincos(subst(f(x)'+(1+tan(x))*f(x)-
   cos(x), f(x)=cos(x)*g(x))))
   ```

   On obtient l'équation que doit vérifier g :

   ```
   cos(x)*(-1+diff(g(x),x)+g(x))
   ```

6. On peut faire faire par `Xcas` le graphe (en rouge) des fonctions f et g qui sont respectivement les solutions exactes de

   $$f(x)' + (1 + \tan(x)) * f(x) - \cos(x) \text{ et } f(0) = 0$$

   et de

   $$g(x)' + g(x) = 1 \text{ et } g(0) = 0$$

   On tape :

   ```
   plotfunc((cos(x)*exp(x)-cos(x))/(exp(x)),
   x,affichage=rouge))
   ```

   et pour comparer avec la solution approchée

   ```
   plotode(cos(x)-(1+tan(x))*y,[x,y],[0,0],plan)
   ```

   On obtient :



On tape :

```
plotfunc((exp(x)-1)/(exp(x)),x,affichage=rouge))
```

et pour comparer avec la solution approchée :

```
plotode(1-y,[x,y],[0,0],plan)
```

On obtient :



3.2 Thème : Outils - calcul vectoriel

– L'exercice proposé au candidat
  On considère un cercle X de centre O et de rayon r et un point P du plan. On pose $d = OP$.

  1. Une droite passant par P coupe le cercle X en A et B . On note E le point du cercle X diamétralement opposé à A.
     Démontrer que $\overrightarrow{PA} \cdot \overrightarrow{PB} = \overrightarrow{PA} \cdot \overrightarrow{PE}$.
     En déduire que $\overrightarrow{PA} \cdot \overrightarrow{PB} = d^2 - r^2$.
  2. Application à l'étude d'une configuration : Dans la figure ci-dessous les droites (AB) et (A_1B_1) sont orthogonales et le point I est le milieu du segment $[AB_1]$. Démontrer que les droites (PI) et (A_1B) sont orthogonales.



– Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Indiquer, pour chacune des questions de l'exercice, les savoirs mis en jeu.
  Q.2) Rédiger une solution de la question 2 de l'exercice telle que le candidat la présenterait à un élève de Première S.
– Une présentation et une solution avec `Xcas` Il faut pour que `Xcas` puisse faire les calculs :
  – soit prendre l'axe des x passant par A,
  – soit prendre 2 points A et B quelconques sur le cercle et un point queconque P sur le segment AB.

En effet, les coordonnées de l'intersection d'une droite et d'un cercle sont souvent compliquées !

1. On fait la figure.
   On tape :

   ```
   assume(a=[1,-5,5,0.1]);
   c:=cercle(0,a):;c;
   A:=point(a);
   assume(p=[0.5,-5,5,0.1]);
   assume(q=[0.3,-5,5,0.1]);
   P:=point(p,q);
   d:=droite(A,P):;d;
   B:=normal(inter(c,d)[0]);
   E:=point(-a);
   L:=inter(c,parallele(P,droite(B,E))):;
   A1:=L[0];
   B1:=L[1];
   d1:=droite(A1,B1):;d1;
   I:=milieu(A,B1);
   ```

   On obtient :

   

2. On utilise comme outil : le calcul vectoriel,
   On tape pour calculer $\overrightarrow{PA} \cdot \overrightarrow{PE}$:

   ```
   normal(dot(vecteur(P,A),vecteur(P,E)))
   ```

   On obtient :

   ```
   -a^2+p^2+q^2
   ```

   On tape pour calculer $\overrightarrow{PA} \cdot \overrightarrow{PB}$:

```
normal(dot(vecteur(P,A),vecteur(P,B)))
```

On obtient :

```
-a^2+p^2+q^2
```

On tape pour savoir si les droites P, I et $A1, B$ sont orthogonales :

```
est_orthogonal(droite(P,I),droite(A1,B))
```

On obtient :

```
1
```

ce qui veut dire `vrai`.
On tape pour montrer que $\overrightarrow{PA} + \overrightarrow{PB1}$ est orthogonale à $\overrightarrow{A1P} + \overrightarrow{PB}$

```
(vecteur(P,A)+vecteur(P,B1))*
(vecteur(A1,P)+vecteur(P,B))
```

On obtient :

```
0
```

On met ainsi en évidence la puissance d'un point par rapport à un cercle

3. On utilise comme outil : les propriétés des angles inscrits.
Soit CD le diamètre passant par P, les triangles PCA et PBD sont semblables car leurs angles sont égaux. Leurs angles C et B sont égaux car ils interceptent le même arc donc :
$\dfrac{PC}{PA} = \dfrac{PB}{PD}$ et on a $\overline{PC} * \overline{PD} = d^2 - r^2$.



Donc $PC * PD = PA * PB$.
Le point P est entre A et B et P, A, B sont alignés donc $\overline{PA}*\overline{PB} = d^2-r^2$

4. On utilise comme outil : les coordonnées,
On tape :

```
l1 :=longueur(P,A)
```

On obtient :

```
sqrt((p-a)^2+q^2)
```

On tape :

```
l2 :=normal(longueur(P,B))
```

On obtient :

```
((a^2-p^2-q^2)*sqrt(q^2+p^2+a^2-2*a*p))/
(a^2-2*a*p+p^2+q^2)
```

On tape :

```
factor(expand(l1*l2))
```

On obtient :

```
-q^2-p^2+a^2
```

3.3 Thème : Suites
Étude du comportement de suites définies par une relation de récurrence du type : $u_{n+1} = f(u_n)$

– L'exercice proposé au candidat

1. On considère la fonction f définie sur [0, 1] par $f(x) = \sqrt{\dfrac{1+x}{2}}$.

   a) Résoudre l'équation $f(x) = x$.

   b) Montrer que pour tout $x \in [0, 1]$, on a $f(x) \geq x$.

2. On considère la suite (u_n) définie par la donnée de son premier terme $u_0 \in [0, 1]$ et par la relation $u_{n+1} = f(u_n)$ pour tout entier naturel n. Montrer que la suite (u_n) converge et déterminer sa limite.

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Énoncer les théorèmes et les outils mis en jeu dans l'exercice.
Q.2) Rédiger un énoncé détaillé de la question 2) pour des élèves de Terminal scientifique.

– Une présentation et une solution avec `Xcas`

1. f est croissante et si $x \in [0; 1]$ on a $\sqrt(2)/2 \geq f(x) \geq 1$ donc $f(x) \in [0; 1]$
   Ainsi si $u_0 \in [0; 1]$, on montre par récurrence que $u_n \in [0; 1]$ pour tout $n \in \mathbb{N}$.
   On tape :

   ```
   f(x) :=sqrt((x+1)/2)
   solve(f(x)=x)
   ```

   On obtient :

   ```
   [1]
   ```

   On tape :

   ```
   g(x) :=f(x)-x
   ```

   On tape :

   ```
   a :=normal(mult_conjugate(diff(g(x))))
   ```

   On obtient :

   ```
   (2*sqrt(2)*x+2*sqrt(2)-sqrt(x+1))/
   ((-2*sqrt(2))*x-2*sqrt(2))
   ```

   On tape :

   ```
   solve(a<0)
   ```

   On obtient :

   ```
   [x>((-7)/8)]
   ```

Donc on a, $g'(x) < 0$ donc $g(x)$ est décroissante de $\sqrt(2)/2$ à 0 sur [0 ;1].
Si on veut simplifier `a`, on peut guider le calcul en multipliant par la quantité conjuguée du numérateur de `a`. On peut faire directement ces opérations en se servant de l'éditeur d'équation qui contient la valeur de `a`.
On tape :

```
b :=mult_conjugate(getNum(a))/getDenom(a)
```

puis

```
normal(getNum(b))/getDenom(b)
```

On obtient :

```
(-8*x^2-15*x-7)/
((2*(-(sqrt(2)))*x+2*(-(sqrt(2)))-sqrt(x+1))*
((-2*sqrt(2))*x-2*sqrt(2)))
```

On tape :

```
plotfunc([x,f(x)])
```

On obtient :



2. Puisque $f(x) \geq x$, on a $u_{n+1} \geq u_n$ donc u est croissante et majorée par 1 donc u est convergente vers l. l vérifie l'équation $l = f(l)$ donc $l = 1$.
On ouvre un écran de géométrie et pour visualiser le comportement de u, on tape :

```
assume(a=[0.1,0,1,0.1])
plotseq(f(x),a)
```

On obtient :


y
1.
1
0.
0.5
1
1.5
2

3.4 Thème : Arithmétique

– L'exercice proposé au candidat
  On appelle diviseur propre d'un entier naturel non nul n, tout diviseur de n qui soit positif et distinct de n. Tout entier naturel non nul égal à la somme de ses diviseurs propres est dit nombre parfait.
  Exemple : 6 est un nombre parfait car il est égal à la somme de ses diviseurs propres soit 1, 2 et 3.
  1. Établir la liste des diviseurs de 28 et 496 et montrer que ce sont deux nombres parfaits.
  2. Vérifier que 28 et 496 sont de la forme $2^n(2^{n+1} - 1)$ où $n \in \mathbb{N}^*$ avec $2^{n+1} - 1$ premier.
  3. Démontrer que pour tout $n \in \mathbb{N}^*$, si $2^{n+1} - 1$ est premier alors $2^n(2^{n+1} - 1)$ est parfait.
  4. Illustrer par un exemple le fait que si $2^{n+1} - 1$ n'est pas premier alors $2^n(2^{n+1} - 1)$ n'est pas parfait.
– Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Quels sont les outils nécessaires à la résolution de l'exercice ?
  Q.2) Rédiger la réponse à la question 3.
– Une présentation et une solution avec `Xcas`
  1. On tape :

     `idivis(28)`

     On obtient :

     `[1,2,4,7,14,28]`

     On fait faire remarquer aux élèves que `ifactor(28)=2^2*7` i.e $28 = 2^2 * 7$. Les diviseurs de 2^2 sont $[1, 2, 4]$ et ceux de 7 sont $[1, 7]$.
     La liste des diviseurs de 28 s'obtient en concaténant $1 * [1, 2, 4] = [1, 2, 4]$ et $7 * [1, 2, 4] = [7, 14, 28]$.
     On tape :

     `idivis(496)`

     On obtient :

     `[1,2,4,8,16,31,62,124,248,496]`

     Car on a : `ifactor(496)=2^4*31`.
     Remarque
     `idivis(2^3*3^2*5)` = `[1,2,4,8,3,6,12,24,9,18,36,72,`
     `5,10,20,40,15,30,60,120,45,90,180,360]`
     De facçon plus générale, si $n = a^\alpha * b^\beta * c^\gamma$ la liste des diviseurs de n est obtenue en concatenant plusieurs listes :
     – On forme la liste $L_0 = [1, a, ..., a^\alpha]$ de longueur $\alpha + 1$ puis,
     – On forme la liste $L_1 = L_0 + b * L_0 + b^2 * L_0 + ...b^\beta * L_0$ de longueur $(\alpha + 1) * (\beta + 1)$ puis,

– On forme la liste des diviseurs L :
$L = L_1 + c * L - 1 + c^2 * L_1, ... c^\gamma * L_1$ de longueur $(\alpha+1)*(\beta+1)*(\gamma+1)$
avec $L_0 = [1, a..a^\alpha]$, $L_1 = L_0 + b * L_0 + b^2 * L_0 + ... b^\beta * L_0$.
Le nombre de diviseurs est donc $(\alpha+1)*(\beta+1)*(\gamma+1)$

2. Montrons que si $(2^{n+1} - 1)$ est premier alors $2^n(2^{n+1} - 1)$ est parfait. Les diviseurs de 2^n sont $L_0 = [1, 2...2^n]$. On tape pour avoir leur somme :

```
sum(2^p,p,0,n)
```

On obtient :

```
2^(n+1)-1
```

Si $2^{n+1} - 1$ est premier, les diviseurs de $2^n(2^{n+1} - 1)$ sont :
$L = L_0 + (2^{n+1} - 1) * L_0$. On tape pour avoir la somme de L :

```
s :=factor(sum(2^p+2^p*(2^(n+1)-1),p,0,n))
```

On obtient :

```
2^(n+1)*(2^(n+1)-1)
```

On tape pour vérifier que sette somme vaut bien $2 * 2^n(2^{n+1} - 1)$:

```
normal(powexpand(s-2*(2^n*(2^(n+1)-1))))
```

On obtient :

```
0
```

3. On considère $n = 3$ on a $2^4 - 1 = 16 - 1 = 15$ n'est pas premier et $2^3 * 15 = 2^3 * 3 * 5 = 120$ n'est pas parfait.
En effet, on tape :

```
L :=idivis(2^3*3*5)
```

On obtient :

```
[1,2,4,8,3,6,12,24,5,10,20,40,15,30,60,120]
```

On tape :

```
sum(L)
```

On obtient :

```
360
```

En effet si $L_0 = [1, 2, 4, 8]$, on a :
$\sum L_0 = 15 = (2^4 - 1)$ mais $\sum 3 * L_0 = 3 * 15$, $\sum(5 * L_0) = 5 * 15$ et $\sum 3 * 5 * L_0 = 15 * 15$.
On a donc $\sum L = (2^4 - 1) * (1 + 3 + 5 + 15) = (2^4 - 1) * 24 = 360$ qui est différent de $(2^4 - 1) * 8 = 120$.
De facçon générale, si $(2^{n+1} - 1)$ n'est pas premier, il existe $a \neq 1$ et $b \neq 1$ tel que $(2^{n+1} - 1) = a * b$.
Donc on a :

$$a * b + 1 = 2^{n+1}$$

On a toujours

$$\sum L_0 = \sum [1, 2, ...2^n] = 2^{n+1} - 1$$

mais, même si a ou b ne sont pas premiers, dans la liste des diviseurs de $2^n(2^{n+1} - 1)$ il y aura :

$$L_0,\ a * L_0,\ b * L_0,\ a * b * L_0$$

On a :

$$1 + a + b + a * b > (a * b + 1) = 2^{n+1} = 2 * 2^n$$

donc, si $(2^{n+1} - 1)$ n'est pas premier $2^n(2^{n+1} - 1)$ n'est pas parfait.

3.5 Thème : Probabilités

– L'exercice proposé au candidat

On s'intéresse à la durée de vie, exprimée en année, d'un téléviseur avant la première panne. On peut modéliser cette situation par une variable aléatoire qui suit une loi de probabilité p de durée de vie sans vieillissement définie sur l'intervalle $[0, +\infty[$. Ainsi la probabilité d'un intervalle $[0, t[$, notée $p([0, t[)$, est la probabilité que le téléviseur tombe en panne avant t année. Cette loi est la loi exponentielle de paramètre λ où λ est un réel strictement positif.

1. Déterminer, en fonction de λ, la valeur de t pour laquelle on a $p([0, t[) = p([t, +\infty[)$.
2. D'après l'étude statistique effectuée par le constructeur, la probabilité que le téléviseur tombe en panne avant la fin de la première année est 0,18. Calculer la valeur exacte de λ.

   Dans la suite de l'exercice, on prendra λ= 0.2.
3. Montrer qu'une valeur approchée de la probabilité que le téléviseur n'ait pas eu de panne au cours des trois premières années, arrondie à 10^{-4} près, est : 0,5488.
4. Sachant que ce téléviseur n'a connu aucune panne au cours des 10 premières années après sa mise en service, quelle est la probabilité qu'il ne connaisse aucune panne au cours des 13 premières années ?
5. Dix téléviseurs neufs de ce type ont été mis en service en même temps. On désigne par X la variable aléatoire égale au nombre de téléviseurs qui n'ont pas eu de panne au cours des trois premières années. Calculer une valeur approchée de la probabilité de l'événement $X = 4$ arrondie à 10^{-4} près.

– Le travail demandé au candidat

Pendant sa préparation, le candidat traitera les questions suivantes :

Q.1) Rédiger une réponse pour chacune des questions 3) et 4) de l'exercice.

Q.2) Commenter l'expression "loi de durée de vie sans vieillissement".

– Une présentation et une solution avec `Xcas`

1. La probabilité $p([0, t[)$ que le téléviseur tombe en panne avant t année suit une loi exponentielle de paramètre λ où λ est un réel strictement positif. On a donc :

$$p([0, t[) = \lambda \sum_{0}^{t} exp(-\lambda x)dx$$

En effet :

$$p([0, +\infty[) = \lambda \sum_{0}^{+\infty} exp(-\lambda x)dx = 1$$

On tape :

```
assume(lambda>0)
```

On obtient :

```
lambda
```

On tape :

```
int(lambda*exp(-lambda*x),x,0,t)
```

On obtient :

```
-exp(-lambda*t)+1
```

On tape :

```
int(lambda*exp(-lambda*x),x,0,inf)
```

Ou on tape :

```
limit(int(lambda*exp(-lambda*x),x,0,t),t,inf)
```

On obtient :

```
1
```

On tape :

```
solve(int(lambda*exp(-lambda*x),x,0,t)=limit
 (int(lambda*exp(-lambda*x),x,t,u),u,inf),t)
```

On obtient :

```
[1/(-lambda)*ln(1/2)]
```

2. On a $p([0, 1[) = 0.18$ c'est à dire :
`int(lambda*exp(-lambda*x),x,0,1)=-exp(-lambda)+1=0.18`
On tape :

```
solve(-exp(-lambda)+1=0.18,lambda)
```

On obtient :

```
0.198450938724
```

3. Dans la suite de l'exercice, on prendra λ= 0,2.
On cherche $1 - p([0, 3[)$ c'est à dire :
`1-int(0.2*exp(-0.2*x),x,0,3)=exp(-0.2*3)`
On tape :

```
exp(-0.2*3)
```

On obtient :

```
0.548811636094
```

Donc la probabilité que le téléviseur n'ait pas eu de panne au cours des trois premières années, arrondie à 10^{-4} près, est : 0,5488.

4. Soit A l'événement : le téléviseur n'a connu aucune panne au cours des 10 premières années après sa mise en service.
   Soit B l'événement : le téléviseur n'a connu aucune panne au cours des 13 premières années après sa mise en service.
   On cherche la probabilité que le téléviseur ne connaisse aucune panne au cours des 13 premières années sachant qu'il n'a connu aucune panne au cours des 10 premières années c'est à dire $P(B/A) = P(A \cap B)/P(A)$.
   On a :
   $P(A) = (1 - p([0, 10[) = \exp(-0.2 * 10)$ et $A \subset B$ donc
   $P(B) = P(A \cap B) = (1 - p([0, 13[)) = \exp(-0.2 * 13)$
   On tape :

   ```
   exp(-0.2*13)/exp(-0.2*10)
   ```

   Ou on tape :

   ```
   exp(-0.2*3)
   ```

   On obtient :

   ```
   0.548811636094
   ```

5. Dix téléviseurs neufs de ce type ont été mis en service en même temps. On cherche, lorsque $k = 4$:

   $$p(X = k) = C_{10}^k \exp(-0.2 * 3 * k) * (1 - \exp(-0.2 * 3)^{10-k}$$

   Pour vérifier que $\sum_{k=0}^{10} p(X = k) = 1$, on tape :

   ```
   sum(comb(10,k)*exp(-0.2*3*k)*
   (1-exp(-0.2*3))^(10-k),k,0,10))
   ```

   On obtient :

   ```
   1
   ```

   On tape :

   ```
   comb(10,4)*exp(-0.2*3*4)*(1-exp(-0.2*3))^6
   ```

   On obtient :

   ```
   0.160716284054
   ```

   La probabilité de l'événement $X = 4$ arrondie à 10^{-4} près est donc 0.1607.

3.6 Thème : Problème de lieu

– L'exercice proposé au candidat :
On considère dans le plan deux droites d et d_1 sécantes en O et de vecteurs directeurs respectifs $\overrightarrow{u}$ et $\overrightarrow{u}'$ tels que $(\overrightarrow{u}, \overrightarrow{u}') = \frac{\pi}{4} (\mathrm{mod} 2\pi)$. On considère deux points A et B situés respectivement sur d et d_1, distincts de O et tels que $OA = OB$. À tout point M du plan on associe la somme notée $s(M)$, des distances du point M aux droites d et d'.

1. Montrer que $s(A) = s(B) = OA\dfrac{\sqrt{2}}{2}$
2. Soit M un point du segment $[AB]$. En utilisant les aires des triangles OMA et OMB, montrer que la somme $s(M)$ est indépendante de la position de M sur le segment $[A, B]$.
3. Calculer la distance OA afin que, pour tout point M du segment $[AB]$, l'on ait $s(M) = 2$.
4. Le point A étant fixé pour satisfaire la condition de la question 3), on note L le lieu des points M du plan tels que $s(M) = 2$. Montrer que L contient un rectangle dont $[AB]$ est un côté.

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Dégager les méthodes et les savoir-faire utilisés dans cet exercice.
Q.2) Présenter une animation sur le module de géométrie dynamique de la calculatrice mettant en évidence le résultat établi dans la question 2) de l'exercice.

– Une présentation et une solution avec `Xcas`

1. On tape :

```
d:=droite(0,1):;d;
d1:=droite(0,1+i):;d1;
O:=point(0);
assume(t=[4,-5,7,0.1]);
A:=point(a);
B:=rotation(0,pi/4,A);
segment(A,B);
assume(t=[0.3,0,1,0.1]);
M:=element(segment(A,B),t);
dist:=longueur(M,d)+longueur(M,d1);
A1:=point(2*sqrt(2));
B1:=rotation(0,pi/4,A1);
M1:=element(segment(A1,B1),t);
s(M):=normal(longueur(M,d))+normal(longueur(M,d1));
normal(s(M1));
```

   On obtient :



On tape :

```
s(A),s(B)
```

On obtient :

```
(sqrt(2))/2*a,(sqrt(2))/2*a
```

2. On tape :

```
aire(triangle(O,A,M))
```

On obtient :

```
(a*t*sqrt(2)*a)/(2*2)
```

On tape :

```
normal(aire(triangle(O,M,B)))
```

On obtient :

```
(-(sqrt(2)))/4*a^2*t+(sqrt(2))/4*a^2
```

On tape :

```
normal(aire(triangle(O,A,M))+aire(triangle(O,M,B))),
          normal(aire(triangle(O,A,B)))
```

On obtient :

```
(sqrt(2))/4*a^2,(sqrt(2))/4*a^2
```

On tape :

```
sd :=normal(2*aire(triangle(O,A,B))/longueur(O,A))
```

On obtient :

```
(sqrt(2))/2*a
```

3. On tape :

```
normal(solve(sqrt(2*x^2)/2=2,x))
```

On obtient :

```
[2*sqrt(2),-2*sqrt(2)]
```

4. On tape :

```
sd(x,y) :=normal(longueur(point(x,y),d)+
        longueur(point(x,y),d1))
```

On obtient :

```
(sqrt(2))/2*abs(x-y)+abs(y)
```

On tape :

```
assume(y>0);assume(X>0);
E1:=normal(subst(subst(sd(x,y)-2,x=X+y),X=x-y));
assume(y>0);assume(X<0);
E2:=normal(subst(subst(sd(x,y)-2,x=X+y),X=x-y));
assume(y<0);assume(X>0);
E3:=normal(subst(subst(sd(x,y)-2,x=X+y),X=x-y));
assume(y<0);assume(X<0);
E4:=normal(subst(subst(sd(x,y)-2,x=X+y),X=x-y));
plotimplicit(E1,x,y);
plotimplicit(E2,x,y);
plotimplicit(E3,x,y);
plotimplicit(E4,x,y);
affichage(droite(y=x),1);
```

On obtient :



3.7 Thème : Analyse Fonctions et équations

- L'exercice proposé au candidat Soit k un réel. On considère la fonction F définie sur $\mathbb{R}$ par :

$$F(x) = \int_0^x e^{kt^2} dt$$

On note C la courbe représentative de F et l'on s'intéresse au nombre de points M_0 d'abscisse x_0 appartenant à C et en lesquels la tangente à C a un coefficient directeur égal à x_0.

1. Montrer qu'un tel point M_0 existe si et seulement si $x_0 > 0$ et vérifie l'équation

$$(E) \qquad lnx = kx^2$$

2. a) En utilisant une calculatrice graphique et en faisant varier les valeurs de k, conjecturer le nombre de solutions de l'équation (E) dans $]0, +\infty[$. b) Si $k > 0$, trouver graphiquement une valeur approchée a de k pour laquelle l'équation (E) a une unique solution dans $]0, +\infty[$.
3. Démontrer que pour $k < 0$, l'équation (E) a une unique solution dans $]0, +\infty[$.

- Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Présenter, à l'aide de la calculatrice, la ou les représentations permettant de faire les conjectures demandées à la question 2).
Q.2) Proposer une solution de la question 3) de l'exercice telle que le candidat la présenterait à des élèves de terminale.
- Une présentation et une solution avec `Xcas`

1. On cherche à résoudre : $F'(x) = x$.
On tape :

```
F(x) :=int(exp(k*t^2),t,0,x)
```

On obtient :

```
(x)->int(exp(k*t^2),t,0,x)
```

On tape :

```
diff(F(x))
```

On obtient :

```
exp(k*x^2)
```

La tangente à C a un coefficient directeur égal à x_0 donc on doit avoir : $exp(k * x_0\hat{}2) = x_0$.
Donc si x_0 existe il est positif puisque égal à une exponentielle.
On tape :

```
ln(exp(k*x^2)=x)
```

On obtient l'équation (E) que doit vérifier x_0 :

```
(k*x^2)=ln(x)
```

2. On ouvre un écran de géométrie, on définit un paramètre k et on trace les courbes $k * x^2$ et $\ln(x)$.
   On tape :

   ```
   assume(k=[0.2,-5,5,0.001])
   ```

   On obtient :

   ```
   parameter(k,-5.0,5.0,0.2,0.1)
   ```

   On tape :

   ```
   plotfunc(k*x^2)
   ```

   On obtient :

   ```
   plotparam(x+(i)*k*x^2,x=-5.0..5.0125)
   ```

   On tape :

   ```
   plotfunc(ln(x))
   ```

   On obtient :

   ```
   plotparam(x+(i)*ln(x),
   x=3.82333054105e-15..5.0125)
   ```

   On obtient pour $k = 0.098$:

   

   Il semble que pour $k \simeq a = 0.184$ les deux courbes sont tangentes. Au point où les 2 courbes sont tangentes on a l'égalité des valeurs des 2 fonctions et de leur derivées donc x et k doivent vérifier :
   $k * x^2 = \ln(x)$ et $2 * k * x = 1/x$ c'est à dire :
   $kx^2 = 1/2$ et $\ln(x) = 1/2 = 0.5$.
   On tape :

```
fsolve(ln(x)=0.5,x)
```

On obtient :

```
1.6487212707
```

On tape :

```
evalf(1/(2*1.6487212707^2)
```

On obtient la valeur approchée a de k qui est solution de (E) :

```
0.183939720586
```

Il y a alors :
- 0 point d'intersection si $k \geq a$
- 1 point d'intersection si $k = a$
- 2 points d'intersection si $0 < k < a$
- 1 point d'intersection si $k \leq 0$

3. Supposons $k < 0$ et montrons qu'alors l'équation (E) a une unique solution dans $]0, +\infty[$. Pour cela, on cherche le signe de :
$g(x) = k * x^2 - \ln(x)$ lorsque $x > 0$.
On tape :

```
assume(k<0);g(x) :=k*x^2-ln(x)
```

On obtient :

```
k, (x)->k*x^2-ln(x)
```

On tape :

```
diff(g(x))
```

On obtient :

```
k*2*x-1/x
```

On tape :

```
assume(x>0);solve(diff(g(x))=0,x)
```

On obtient :

```
[]
```

On tape :

```
solve(diff(g(x))<0,x)
```

On obtient :

```
[x>0]
```

Pour $x > 0$ on a $g'(x) < 0$ donc $g(x)$ est décroissante.
On tape :

```
limit(g(x),x,0)
```

On obtient :

```
+infinity
```

On tape :

```
limit(g(x),x,+infinity)
```

On obtient :

```
-infinity
```

Donc $g(x)$ ne s'annule qu'en un point lorsque $k<0$.
Par exemple pour $k=-1$ on tape :

```
k :=-1;plotfunc(g(x))
```

On obtient :



On tape :

```
k :=-1;fsolve(g(x)=0,x,1,newton_solver)
```

On obtient :

```
0.652918640419
```

On tape :

```
g(0.652918640419),g(0.652918640420)
```

On obtient :

```
6.11066752754e-13,-2.20268248086e-12
```

Donc pour $k=-1$ on a :

$$0.652918640419 < x_0 < 0.652918640420$$

.

3.8 Thème : Problèmes d'incidence

- L'exercice proposé au candidat :
  Soit un parallélogramme $ABCD$ situé dans un plan P et soit S un point de l'espace n'appartenant pas à P . On note respectivement I, J et K les milieux des segments $[SA]$, $[SB]$ et $[SC]$.
- a) Montrer que les plans P et (IJK) sont parallèles.
  b) Montrer que le plan (IJK) coupe $[SD]$ en son milieu.
- Quelle est l'intersection des plans (CIJ) et P ?
- En déduire l'intersection des plans (CIJ) et (SAD).



- Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Énoncer les théorèmes mis en jeu dans l'exercice.
  Q.2) Présenter un corrigé de la question 1) pouvant être présenté à une classe de lycée.
- Une présentation et une solution avec `Xcas`
  1. On tape :

     ```
     A:=point([0,0,0]);
     ```

```
B:=point([0,4,0]);
D:=point([4,-1,0]);
parallelogramme(A,B,D,C);
S:=point(2,2,4);
I:=milieu(A,S);
J:=milieu(B,S);
K:=milieu(C,S);
polyedre(A,B,C,D,S)
```

On obtient :

`La figure du début`

On tape :

`est_parallele(plan(A,B,C),plan(I,J,K))`

On obtient :

`1`

2. On tape :

`longueur2(M,S)-longueur2(M,D)`

On obtient :

`0`

Ou on tape :

`est_isocele(M,S,D)`

On obtient :

`1`

3. On tape :

```
d :=inter_unique(plan(C,I,J),plan(A,B,C)) :;
affichage(d,2)
```

On obtient : `le tracé de d en vert`
On tape :

`est_element(D,d)`

On obtient :

`1`

Ou on tape :

`d==droite(C,D)`

On obtient :

`1`

4. On tape :

```
d1 :=inter_unique(plan(C,I,J),plan(S,A,D)) :;
                affichage(d1,1)
```

On obtient la figure avec `le tracé de d1 en rouge` :



On tape :

```
d1==droite(D,I)
```

On obtient :

```
1
```

Ou on tape :

```
est_element(D,d1)
```

On obtient :

```
1
```

3.9 Thème : Divers types de raisonnements (par l'absurde, par récurrence...)

– L'exercice proposé au candidat
  Étant donné un entier naturel $n \geq 2$, on se propose détudier l'existence de trois entiers naturels x, y et z tels que $x^2 + y^2 + z^2 = -1 \pmod{2^n}$.

  1. Cas où $n = 2$: Montrer que 1, 3 et 5 sont solutions du problème.
  2. On suppose dorénavant que n est un entier naturel et que $n \geq 3$.
     Supposons qu'il existe trois entiers naturels x, y et z tels que :
     $x^2 + y^2 + z^2 = -1 \pmod{2^n}$.

     a) Montrer que les entiers x, y et z sont tous impairs ou que deux d'entre eux sont pairs.

     b) On suppose que x et y sont pairs et que z est impair. Montrer qu'on a alors $x^2 + y^2 + z^2 = 1 \pmod 4$ et en déduire une contradiction.

     c) On suppose que x, y et z sont impairs.
     Montrer qu'on a $x^2 + y^2 + z^2 = 3 \pmod 8$ et conclure.

– Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Énoncer les théorèmes et les outils mis en jeu dans l'exercice.
  Q.2) Présenter une correction détaillée de la question 2.c telle que le candidat la proposerait à des élèves de Terminale S.

– Une présentation et une solution avec `Xcas`

  1. On tape :

     ```
     (1^2+3^2+5^2)% (2^2)
     ```

     On obtient :

     ```
     -1 % 4
     ```

  2. – Si $x^2 + y^2 + z^2 = -1 \pmod{2^n}$ cela entraine que $x^2 + y^2 + z^2$ est un nombre impair c'est a dire la somme de 2 nombres pairs et d'un nombre impair ou de 3 nombres impairs.
     On tape :

     ```
     ((2*p+1)^2+(2*q+1)^2+(2*r+1)^2)% 4
     ```

     On obtient :

     ```
     -1 % 4
     ```

     Donc pour $n = 2$, tous les triplets de nombres impaires sont des solutions. On tape :

     ```
     ((2*p)^2+(2*q+1)^2+(2*r+1)^2)%4
     ```

     On obtient :

     ```
     2 % 4
     ```

     On tape :

     ```
     ((2*p)^2+(2*q)^2+(2*r+1)^2)%4
     ```

     On obtient :

     ```
     1 % 4
     ```

     On tape :

     ```
     ((2*p)^2+(2*q)^2+(2*r)^2)%4
     ```

On obtient :

```
0 % 4
```

Donc pour $n = 2$, tous les triplets de nombres impairs sont des solutions et ce sont les seules.

- Si $x^2 + y^2 + z^2 = -1 \pmod{2^n}$ avec $n \geq 3$ cela veut dire :
  il existe $k \in \mathbb{N}$ tel que : $x^2 + y^2 + z^2 = 2^n * k - 1$ ou encore
  $x^2 + y^2 + z^2 = 4 * 2^{n-2} * k - 1$ avec $n - 2 \geq 1$ donc $x^2 + y^2 + z^2 = -1 \pmod 4$ donc si l'équation $x^2 + y^2 + z^2 = -1 \pmod{2^n}$ avec $n \geq 3$ a une solution alors $x^2 + y^2 + z^2 = -1 \pmod 4$ donc d'après les résultats ci-dessus, les 3 nombres x, y, z sont impairs.
- On cherche tout d'abord $(2 * r + 1)^2 \pmod 8$.
  On a soit r pair ($r = 2 * n$), soit r impair ($r = 2 * n + 1$) donc on tape :

  ```
  (4*n+1)^2 %8, (4*n+3)^2 %8
  ```

  On obtient :

  ```
  1%8, 1%8
  ```

  Donc $(2 * r + 1)^2 = 1 \pmod 8$
  Donc $(2 * p + 1)^2 + (2 * q + 1)^2 + (2 * r + 1)^2) = 3 \pmod 8$.
  On a montré que si l'équation $x^2 + y^2 + z^2 = -1 \pmod{2^n}$ a une solution alors :

  $$x^2 + y^2 + z^2 = 3 \pmod 8$$

  Si $x^2 + y^2 + z^2 = -1 \pmod{2^n}$ avec $n \geq 3$, il existe $k \in \mathbb{N}$ tel que :
  $x^2 + y^2 + z^2 = 2^n * k - 1$ ou encore
  $x^2 + y^2 + z^2 = 8 * 2^{n-3} * k - 1$ avec $n - 3 \geq 0$ donc $x^2 + y^2 + z^2 = -1 \pmod 8$.
  On a donc une contradiction.
  Conclusion
  Pour $n > 2$, l'équation $x^2+y^2+z^2 = -1 \pmod{2^n}$ n'a pas de solution.

3.10 Thème : Problèmes de calcul de grandeurs Calculs de longueurs, d'aires et de volumes

– L'exercice proposé au candidat
Dans la figure ci-dessous le triangle ABC est équilatéral et les droites (d), (d_1) et (d_2) sont des droites parallèles passant respectivement par les sommets C, A et B.
On note a la distance de (d) à (d_1) et b celle de (d) à (d_2) ; on se propose de calculer, en fonction de a et b, l'aire du triangle ABC.



1. Le cercle circonscrit à ABC recoupe la droite (d) en un point P.
Montrer que $AP = \dfrac{2a}{\sqrt{3}}$ et que $BP = \dfrac{2b}{\sqrt{3}}$.
2. En déduire que $AB^2 = \dfrac{4(a^2 + b^2 + ab)}{3}$
3. Calculer l'aire du triangle ABC en fonction de a et b.

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Dégager les méthodes et les savoirs mis en jeu dans la résolution de l'exercice.
Q.2) Proposer la rédaction d'une solution à la question 1).
– Une présentation et une solution avec `Xcas`
Pour faire la figure avec `Xcas`, il faut connaitre les paramètres qui définissent la figure. Ici, les paramètres sont a et b. Mais comment faire cette figure lorsqu'on connait a et b ? On voit que le problème est assez mal posé car on ne sait pas si

la droite (d) se trouve toujours entre (d_1) et d_2 et si le triangle ABC est de sens direct.
Si on se donne un rep'ere avec le point C comme origine et (d) comme axe des x. On trace facilement les droites (d_1) et (d_2) de part et d'autre de (d) si on suppose que (d) se trouve toujours entre (d_1) et $d_2)$ et que (d_1) a pour équation $y = a$ et (d_2) a pour équation $y = -b$.
Comment maintenant construire le triangle équilatèral ABC direct pour que A soit sur $(d_1$ et B soit sur $(d_2$. Soit $(d_{11}$ la droite image de d_1 par la rotation de centre C et d'angle $\frac{\pi}{3}$. $(d_{11}$ coupe d_2 en B. Puis A est le troisième sommet du triangle équilatéral ABC.

1. On tape pour faire la figure :

   ```
   C:=point(0);
   assume(a=[3,0,5,0.1]);
   assume(b=[1.0,0,5,0.1]);
   d1:=droite(y=a);
   d11:=rotation(0,pi/3,d1,affichage=1);
   d2:=droite(y=-b);
   B:=inter_unique(d11,d2);
   A:=rotation(0,-pi/3,B);
   c:=circonscrit(A,B,C);
   P:=inter(c,droite(y=0))[1];
   ```

   On obtient :

   

2. On tape :

```
simplify(longueur(P,A))
```

On obtient :

```
(2*a*sqrt(3))/3
```

On tape :

```
simplify(longueur(P,B))
```

On obtient :

```
(2*b*sqrt(3))/3
```

on peut aussi simplement remarquer que $\sin(\pi/3) = a/AP = \sqrt{3}/2$ et que $\sin(\pi/3) = b/BP = \sqrt{3}/2$ d'où les résultats.

3. On tape :

```
simplify(longueur2(A,B))
```

On obtient :

```
(4*a^2+4*a*b+4*b^2)/3
```

On peut calculer cette longueur de plusieurs façons :

– Dans le triangle ABP l'angle P vaut $2\pi/3$ car dans le quadrilatère inscriptiple $APBC$ on a :
  $\widehat{P} + \widehat{C} = \widehat{P} + \pi/3 = \pi$.
  On utilise alors, dans le triangle ABP d'angle P égal la relation :
  $$AB^2 = AP^2 + BP^2 - 2AP * BP\cos(\frac{2\pi}{3}) = AP^2 + BP^2 + AP * BP$$
– On utilise la trigonométrie on note $c = \rightarrow Bx, \rightarrow BC$ et on a : $sin(c) = b/BC = b/AB$ et $sin(\pi/3 + c) = (a + b)/AB = sin(c)/2 + cos(c) * \sqrt{3}/2$
  donc $cos(c) * \sqrt{3} = (2 * a + b)/AB$ et
  $3\cos(c)^2 = 3 - 3b^2/AB^2 = (2a + b)^2/AB^2$
  On obtient donc :
  $3AB^2 = (2a + b)^2 + 3b^2 = 4(a^2 + b^2 + ab$
– On calcule les coordonnèes de B dans le repère d'origine C.
  B est sur la droite $y = -b$ et sur la droite transformée de Ax par la rotation de sommet C et d'angle $\pi/3$. Cette droite a pour équation $y = \sqrt{3}x + 2a$ car elle a comme pente $\tan(\pi/3) = \sqrt{3}$ et passe par le point $a*\exp(i*(\pi/2+\pi/3)) = a*(-\sqrt{3}+i)/2$ transformé de $a\exp(i*\pi/2)$.

4. On tape :

```
simplify(aire(A,B,C))
```

On obtient :

```
(a^2*sqrt(3)+a*b*sqrt(3)+b^2*sqrt(3))/3
```

5. Pour comprendre les calculs fait par `Xcas`.
On peut chercher les abscisses de A et de B.
On tape :

```
aa :=normal(abscisse(A))
```

On obtient :

```
(-(sqrt(3)))/3*a+(-2*sqrt(3))/3*b
```

On tape :

```
bb :=normal(abscisse(B))
```

On obtient :

```
((-2*sqrt(3))*a)/3+((-(sqrt(3)))*b)/3
```

Pour faire le détail des calculs de l'abscisse de A et de B.
On tape :

```
k :=exp(i*pi/3)*(t+i*a)
```

On obtient k l'affixe de B en fonction de t l'abscisse de A :

```
(1/2+((i)*sqrt(3))/2)*(t+(i)*a)
```

On tape :

```
re(k),im(k)
```

On obtient :

```
t/2-(sqrt(3)*a)/2,a/2+(t*sqrt(3))/2
```

On tape puisque l'ordonnée de B est $-b$:

```
solve(im(k)=-b,t)
```

On obtient l'abscisse de A :

```
[1/(sqrt(3))*(-a-2*b)]
```

On tape :

```
normal(subst(re(k),t=1/(sqrt(3))*(-a-2*b)))
```

On obtient l'abscisse de B :

```
(-2*sqrt(3))/3*a+(-(sqrt(3)))/3*b
```

On tape (Attention l'ordonnee de B est $-b$!) :

```
normal((aa-bb)^2+(a+b)^2)
```

On obtient la valeur de AB^2 :

```
(4*a^2+4*a*b+4*b^2)/3
```

On tape pour avoir l'aire de ABC qui est $AB^2 * sqrt(3)/4$:

```
normal(((aa-bb)^2+(a+b)^2)*sqrt(3)/4)
```

On obtient :

```
(sqrt(3)*aˆ2+sqrt(3)*a*b+sqrt(3)*bˆ2)/3
```

Puisque la médiatrice du segment PC passe par le centre G du cercle circonscrit et que G est l'isobarycentre de A, B, C on tape :

```
G :=isobarycentre(A,B,C) :;
```

On tape :

```
simplify((abscisse(A)+abscisse(B)+abscisse(C))/3);
```

On obtient l'abscisse de G :

```
(-a*sqrt(3)-b*sqrt(3))/3
```

On vérifie en tapant :

```
simplify(abscisse(G));
```

On tape :

```
simplify(2*abscisse(G));
```

On obtient l'abscisse de P :

```
(-2*a*sqrt(3)-2*b*sqrt(3))/3
```

On vérifie en tapant :

```
simplify(abscisse(P))
```

– Une solution en `géométrie`
  On suppose que la figure a été réalisée :



1. Puisque A, B, C, P sont cocycliques et que P et B sont sur le même arc $\overset{\frown}{AC}$, on a :

$$\widehat{APC} = \widehat{ABC} = \frac{\pi}{3}$$

car ces angles interceptent l'arc $\overset{\frown}{AC}$.
Puisque A, B, C, P sont cocycliques et que P et A sont sur le même arc $\overset{\frown}{BC}$, on a :

$$\widehat{BPC} = \widehat{BAC} = \frac{\pi}{3}$$

car ces angles interceptent l'arc $\overset{\frown}{BC}$.
donc on a :

$$\sin(\widehat{APC}) = \sin(\frac{\pi}{3}) = \frac{\sqrt{3}}{2} = \frac{a}{AP}$$

$$\sin(\widehat{BPC}) = \sin(\frac{\pi}{3}) = \frac{\sqrt{3}}{2} = \frac{b}{BP}$$

On a donc :

$$AP = \frac{2a}{\sqrt{3}} \text{ et } BP = \frac{2b}{\sqrt{3}}$$

2. Pour calculer AB^2, on considère le triangle APB donc l'angle P vaut $\frac{2\pi}{3}$ donc on a :

$$AB^2 = AP^2 + BP^2 - 2AP * BP * \cos(\frac{2\pi}{3}) = \frac{4(a^2 + b^2 + ab)}{3}$$

3. L'aire d'un triangle équilatéral de côté l vaut : $l^2 * \frac{\sqrt{3}}{4}$ donc l'aire du triangle ABC vaut :

$$\frac{\sqrt{3}(a^2 + b^2 + ab)}{3}$$

3.11 Thème : Intégration. Calcul d'intégrales par des méthodes variées

– L'exercice proposé au candidat
  On considère la fonction F définie sur $[0, +\infty[$ par

$$F(x) = \int_0^x \frac{1}{\sqrt{1+t^2}} dt$$

  1. Montrer que F est continue et strictement croissante sur $[0, +\infty[$.
     On considère la fonction u définie sur $[0, +\infty[$ par :

$$u(x) = \frac{e^x - e^{-x}}{2}$$

  2. a) Calculer la dérivée de la fonction $F \circ u$.
     b) En déduire que, pour tout réel $x \in [0, +\infty[$, on a $F \circ u(x) = x$.
     c) Calculer $F(2)$.

– Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Préciser les théorèmes utilisés dans cet exercice.
  Q.2) Proposer une solution de la question 2) telle que le candidat la présenterait à une classe.
– Une présentation et une solution avec `Xcas`
  1. On tape pour définir $F(x)$:

```
F(x) :=int(1/sqrt(1+t^2),t,0,x)
```

     On tape pour avoir $F'(x)$:

```
normal(diff(F(x),x))
```

     On obtient puisque $F(x)$ est la primitive de $1/\sqrt{1+x^2}$ qui s'annule en 0 :

```
(sqrt(x^2+1))/(x^2+1)
```

  2. On tape pour définir $u(x)$, $H(x) = F(u(x))$ et $H'(x)$:

```
u(x) :=(exp(x)-exp(-x))/2
H(x) :=int(1/sqrt(1+t^2),t,0,u(x))
tsimplify(diff(H(x),x))
```

     On obtient :

```
1
```

     En effet on peut faire le calcul à la main :
     $H'(x) = u'(x)F'(u(x)) = u'(x)/\sqrt{1+u(x)^2}$ donc
     $1+u(x)^2 = (4+\exp(2x)+\exp(-2x)-2)/4 = (\exp(x)+\exp(-x))^2/4$
     On a :

$u'(x) = (\exp(x) + \exp(-x))/2$ donc :
$H'(x) = 1$
Pour tout réel $x \in [0, +\infty[$, on a $H'(x) = 1$ et $H(0) = 0$ donc on a :

$$H(x) = F \circ u(x) = x, \text{ si } x \in [0, +\infty[$$

3. On tape :

```
solve(u(x)=2,x)
```

On obtient :

```
[ln(sqrt(5)+2)]
```

On en déduit que :
$F(2) = F(u(\ln(\sqrt{5} + 2))) = H(\ln(\sqrt{5} + 2)) = \ln(\sqrt{5} + 2)$.
On tape pour vérifier :

```
F(2)
```

On obtient :

```
-(ln(sqrt(5)-2))
```

et on a bien $-(\ln(\sqrt{5} - 2)) = \ln(\sqrt{5} + 2)$.
On tape pour calculer $F(x)$:

```
F(x)
```

On obtient :

```
-(ln(sqrt(x^2+1)-x))
```

ou encore, on tape :

```
normal(lncollect(-(ln(sqrt(x^2+1)-x))-
        ln(sqrt(x^2+1)+x)))
```

On obtient :

```
0
```

Donc
$$F(x) = \ln(\sqrt{x^2 + 1} + x)$$
On peut aussi remarquer que $F(x)$ et $\ln(\sqrt{x^2 + 1} + x)$ ont la même dérivée et que $F(0) = \ln(\sqrt{0^2 + 1} + 0)) = 0$.
On a donc :
$F(u(x)) = \ln(\sqrt{1 + u(x)^2} + u(x)) =$
$\ln((\exp(x) + \exp(-x))/2 + (\exp(x) - \exp(-x))/2) = \ln(\exp(x))$ Donc
$H(x) = F(u(x)) = x$

3.12 Thème : Géométrie. Interprétation géométrique des nombres complexes

– L'exercice proposé au candidat
Le plan complexe est rapporté à un repère orthonormal direct $(O, \overrightarrow{u}, \overrightarrow{v})$. On note (c) le cercle de centre O et de rayon 1.

1. Faire la figure
2. Soit A un point de (c) d'affixe a. On note (T_a) la tangente en A à (c).
   Soit M un point du plan d'affixe z.
   Montrer que M appartient à (T_a) si et seulement si $\dfrac{z-a}{a}$ est imaginaire pur.
3. Déduire que M appartient à (T_a) si et seulement si z vérifie l'égalité :
   $z\overline{a} + \overline{z}a = 2$.
4. Soient A d'affixe a et B d'affixe b deux points distincts de (c) tels que $a + b \neq 0$. Montrer que les droites (T_a) et (T_b), tangentes à (c) respectivement en A et B, sont sécantes et que leur point d'intersection a pour affixe $\dfrac{2ab}{a+b}$.

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Dégager les méthodes et les savoirs mis en jeu dans la résolution de l'exercice.
Q.2) Proposer une solution de la question 1) telle que le candidat la présenterait à une classe.

– Une présentation et une solution avec `Xcas`

1. La figure
   On remarque que pour faire la figure il faut choisir comme paramètre l'argument t_a de l'affixe de A. On a alors $a = \exp(i * t_a)$.
   On tape :

```
c:=cercle(0,1):;c;
assume(ta=[0.4,-3.2,3.2,0.1]);
A:=point(exp(i*ta));
a:=affixe(A);
Ta:=tangente(c,A):;Ta;
assume(tt=[0.7,-5,5,0.1]);
M:=element(Ta,tt);
m:=affixe(M);
N:=point(a+a*i*tt);
assume(tb=[1.3,-3.2,3.2,0.1]);
B:=point(exp(i*tb);
b:=affixe(B);
Tb:=tangente(c,B):;Tb;
K:=inter_unique(Ta,Tb);
```

   On obtient :



Dans la figure ci-dessus, le point M sur la tangente T_a et le point N est un point d'affixe $n = a + a * i * tt$ c'est à dire tel que $\dfrac{n-a}{a}$ est imaginaire pur.

2. On tape :

```
simplify(re((m-a)/a))
```

On obtient :

```
0
```

Donc si M d'affixe m est sur la tangente T_a, on a `re((m-a)/a)=0` c'est à dire $\dfrac{m-a}{a}$ est imaginaire pur.
On tape :

```
est_element(N,Ta)
```

On obtient :

```
1
```

Donc si N d'affixe n tel que $\dfrac{n-a}{a}$ est imaginaire pur, alors N est un élément de T_a.

3. On tape :

```
simplify(m*conj(a)+conj(m)*a)
```

On obtient :

```
2
```

Donc si M d'affixe m est sur la tangente T_a, on a $m\overline{a} + a\overline{m} = 2$.
On tape :

```
L :=csolve(r*conj(a)+conj(r)*a=2,r)
```

On obtient (`‘ x‘` désigne la partie réelle de r) :

```
[‘ x‘+(i)*1/(sin(ta))*(-‘ x‘*cos(ta)+1)]
```

On tape :

```
simplify(re((L[0]-a)/a))
```

On obtient :

```
0
```

Donc si R d'affixe r vérifie $r * \overline{a} + \overline{r} * a = 2$, on a $(r - a)/a$ est imaginaire pur donc d'après la question précédente R est sur la tangente T_a.

4. Les tangentes à un cercle sont concourantes si et seulement si leurs points de contact ne sont pas diamétralement opposés, c'est à dire ici, si $a+b \neq 0$.
   On tape :

```
simplify(trig2exp(affixe(K)))
```

On obtient :

```
((2*i)*exp((i)*ta)*exp((i)*tb))/
((i)*exp((i)*ta)+(i)*exp((i)*tb))
```

On tape (on quote `a` et `b` pour avoir le résultat en fonction de `a` et `b`) :

```
subst(k,[exp(i*ta)=’a’,exp(i*tb)=’b’])
```

Ou on tape :

```
subst(k,[exp(i*ta)=quote(a),exp(i*tb)=quote(b)])
```

On obtient :

```
((2*i)*a*b)/((i)*a+(i)*b)
```

Donc l'affixe k de K vaut :

$$\frac{2ab}{a+b}$$

3.13 Thème : Probabilités

– L'exercice proposé au candidat Cet exercice est un QCM. Pour chaque affirmation une seule des réponses A, B ou C est exacte et il s'agit de la trouver.

1. Dans une classe de 31 élèves, 12 élèves jouent au tennis, 8 élèves jouent au football et 5 élèves jouent à la fois au tennis et au football. On interroge au hasard un élève de cette classe. La probabilité que cet élève ne joue ni au tennis ni au football est :

$$\boxed{A}\ \frac{6}{31} \qquad \boxed{B}\ \frac{16}{31} \qquad \boxed{C}\ \frac{11}{31}$$

2. Une urne contient trois boules blanches et deux boules noires. On tire successivement et au hasard deux boules en respectant le protocole suivant : si la première boule tirée est noire alors on la remet dans l'urne avant de tirer la seconde boule. Si la première boule est blanche alors on ne la remet pas dans l'urne avant de tirer la seconde boule.

   La probabilité d'obtenir exactement une boule blanche à l'issue des deux tirages est :

$$\boxed{A}\ \frac{3}{5} \qquad \boxed{B}\ \frac{27}{50} \qquad \boxed{C}\ \frac{12}{22}$$

3. On dispose d'une urne U_1 contenant quatre jetons numérotés 1, 1, 2, 3 et d'une urne U_2 contenant trois jetons numérotés 2, 3, 3. On tire au hasard un jeton dans chaque urne et on appelle X la variable aléatoire qui, à chaque tirage de deux jetons, associe la valeur absolue de la différence des numéros portés par les deux jetons. L'espérance mathématique de X est :

$$\boxed{A}\ 1 \qquad \boxed{B}\ \frac{13}{12} \qquad \boxed{C}\ \frac{5}{6}$$

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Indiquer pour chacun des items de ce QCM les savoirs mis en jeu pour trouver la réponse exacte.
Q.2) Justifier la réponse à la question 2).

– Une présentation et une solution avec ou sans `Xcas`

1. On a donc 12-5=7 elèves ne jouent qu'au tennis, 8-5=3 ne jouent qu'au football,5 jouent au tennis et au foot et 31-(7+3+5)=16 elèves ne jouent ni au tennis ni au foot. La reponse est donc

```
16/31
```

2. On peut avoir :
   une boule blanche au premier tirage puis une noire (probabilité=3/5*2/4)
   ou une boule noire au premier tirage puis une blanche (probabilité=2/5*3/5)
   On tape :

```
3/5*2/4+2/5*3/5
```

On obtient :

```
27/50
```

3. La variable X peut prendre comme valeur :
   0 pour le tirage (2,2) ou (3,3) ou (3,3)
   1 pour le tirage (1,2),ou (1,2) ou (2,3) ou (2,3) ou (3,2)
   2 pour le tirage (1,3) ou (1,3) ou (1,3) ou (1,3) Donc :
   $P(X = 0) = 3/12$
   $P(X = 1) = 5/12$
   $P(X = 2) = 4/12$
   $E(X) = 5/12 + 2 * 4/12$
   On tape :

   ```
   5/12+2*4/12
   ```

   On obtient :

   ```
   13/12
   ```

3.14 Thème : Problèmes sur les configurations

– L'exercice proposé au candidat
Sur la figure ci-dessous :



– les points A, O_1, O_2 et O_3 sont alignés dans cet ordre ;
– les cercles $(c_1), (c_2)$ et (c_3) ont pour centres respectifs O_1, O_2, O_3 et pour rayons respectifs 10, 20 et 59 millimètres ;
– le cercle (c_2) est tangent aux cercles (c_1) et (c_3) ;
– le point A appartient à (c_1) ;

1. Construire, sur la figure jointe, à la règle et au compas, une droite (d) passant par A et tangente à (c_3) (on laissera visibles les traits de construction).
2. On appelle H le projeté orthogonal de O_2 sur (d). Calculer la distance O_2H et justifier que (d) coupe (c_2) en deux points, que l'on notera B et C.
3. Calculer la distance BC.

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Préciser les savoirs mis en jeu dans la résolution de l'exercice.
Q.2) Proposer une solution de la question 2) telle que le candidat la présenterait à une classe.
– Une présentation et une solution avec `Xcas`

1. Si (d) passe par A et est tangente a c_3 au point M, on a l'angle $\widehat{AMO_3}$ est droit donc M se trouve sur le cercle c de diamètre AO_3. On fait la figure.
On tape :

```
O1:=point(-4);
c1:=cercle(O1,1);
O2:=point(-1);
c2:=cercle(O2,2);
O3:=point(69/10);
c3:=cercle(O3,59/10);
A:=point(-5);
c:=cercle(A,O3,affichage=1);
L:=inter(c,c3):;
M:=L[0];d:=droite(A,M);
N:=L[1];droite(A,N);
H:=projection(d,O2);
LL:=inter(c2,d):;
B:=affichage(LL[0],quadrant4);
C:=affichage(LL[1],quadrant3);
segment(O2,H,affichage=1);
segment(O3,M,affichage=1);
```

On obtient :



2. Pour calculer O_2H on remarque que les triangles AO_2H etAO_3M sont homothétiques, donc $O_2H/O_3M = AO_2/AO_3 = 40/(20+40+59)$ On tape :

```
59/10*4/(2+4+59/10)
```

On obtient :

```
236/119
```

On vérifie :

```
normal(longueur(O2,H))
```

On obtient :

```
236/119
```

3. On a $(BC/2)^2 = O_2B - O_2H^2 = 4 - (236/119)^2$.
   On tape :

   ```
   normal(2*sqrt(4-(236/119)^2))
   ```

   On obtient :

   ```
   (4*sqrt(237))/119
   ```

   On vérifie :

   ```
   normal(longueur(B,C))
   ```

   On obtient :

   ```
   (4*sqrt(237))/119
   ```

3.15 Thème : Intégration

- L'exercice proposé au candidat L'exercice a pour objet d'étudier la suite $(I_n)_{n\in\mathbb{N}^*}$ définie pour tout entier $n \geq 1$ par :

$$I_n = \int_1^e x^2(\ln x)^n dx$$

  1. Calculer I_1 et montrer que pour tout entier $n \geq 1$, on a :

$$I_{n+1} = \frac{e^3}{3} - \frac{n+1}{3} I_n$$

  2. a) À l'aide d'une calculatrice, donner une conjecture sur le sens de variation et la convergence de la suite $(I_n)_{n\in\mathbb{N}^*}$.
     b) Démontrer les propriétés conjecturées à la question 2) a).
- Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Indiquer, pour chaque question de l'exercice, les savoirs mis en jeu.
  Q.2) Présenter une solution de la question 2).
- Une présentation et une solution avec `Xcas`
  1. On tape :

```
I(n) :=int(x^2*ln(x)^n,x,1,e)
normal(I(1))
```

     On obtient :

```
2/9*exp(1)^3+1/9
```

  2. On tape :

```
factor(ibpu(x^2*ln(x)^(n+1),ln(x)^(n+1)))
```

     On obtient :

```
[(x^3*ln(x)^(n+1))/3,
((-(x^2))*(n+1)*ln(x)^(n+1))/(3*ln(x))]
```

     On tape :

```
assume(n>0);preval(x^3*ln(x)^(n+1)/3,1,e)
```

     On obtient :

```
n,1/3*exp(1)^3
```

     Donc

$$I_{n+1} = \frac{e^3}{3} - \frac{n+1}{3} I_n$$

3. Si $x \in [0; e]$, on a $0 \leq \ln(x)^{n+1} \leq \ln(x)^n$ donc $0 < I_{n+1} < I_n$.
La suite I_n est donc positive et décroissante.
On a $I_{n+1} > 0$, donc en se servant de la relation de récurrence trouvée, on a $I_n < \dfrac{e^3}{n+1}$.
On peut aussi utiliser l'inégalité :
si $x \in [1; e]$ alors $1/x \leq x^2 = x^3/x \leq e^3/x$ donc

$$\frac{1}{n+1} = \int_1^e \frac{\ln(x)^n}{x} dx \leq I_n \leq \int_1^e e^3 * \frac{\ln(x)^n}{x} dx = \frac{e^3}{n+1}$$

On peut aussi dire : la suite I_n est décroissante et minorée donc est convergente vers l. l est forcément égale à 0 car on doit avoir :
$\lim((n+1) * I_n) = e^3/3$ quand $n- > +\infty$.
Donc I_n tend vers 0 quand n tend vers l'infini.
On peut avoir un encadrement de I_n puisque :
$I_{n+1} < \dfrac{e^3}{n+2}$ ce qui entraine $\dfrac{e^3}{n+1}(1 - \dfrac{3}{n+2}) < I_n$, donc

$$\frac{e^3}{n+1}\frac{n-1}{n+2} < I_n < \frac{e^3}{n+1}$$

On peut raffiner car $I_{n+1} > \dfrac{e^3}{n+2}\dfrac{n}{n+3}$ donc :
$\dfrac{(n-1)e^3}{(n+1)(n+2)} < I_n < \dfrac{e^3}{n+1} - \dfrac{3ne^3}{(n+1)(n+2)(n+3)}$ d'où

$$\frac{(n-1)e^3}{(n+1)(n+2)} < I_n < \frac{e^3(n^2+2n+6)}{(n+1)(n+2)(n+3)}$$

On tape par exemple avec Digits :=20 :
`evalf(e^3*48/(50*51))` renvoie :
`0.37808069502470903982 4`
`evalf(e^3*(49^2+2*49+6)/(50*51*52))` renvoie :
`0.37944396676157698107 4`
donc

$$0.378080695024709039824 < I_{49} < 0.379443966761576981074$$

Si on tape avec Digits :=20 :
`romberg(x^2*ln(x)^49,x,1,exp(1))`
On obtient :
`0.379370865517700976837`
Si on tape :
`evalf(I(49), Digits :=20)`
On obtient :
`0.18446744073709551616e20`
c'est tres mauvais ! ! ! !

Si on tape :

`evalf(I(49), Digits :=40)`

On obtient :

`0.37500000000000000000000000000000000000000`

c'est encore mauvais ! !

Si on tape :

`evalf(I(49), Digits :=50)`

On obtient :

`0.37937086551391985267400741577148437500000000000000`

c'est correct !

Si on tape :

`I(49)`

On obtient

$$-\frac{96505889631974913501432990802927297808827257731017 4}{22876792454961}*\exp(1)^3-$$

$$\frac{-19383726095081060488362942943011041297039360000000000}{22876792454961}$$

3.16 Thème : Géométrie Problèmes de recherche de lieux géométriques

- L'exercice proposé au candidat Soient A, B et C trois points non alignés. On se propose de déterminer l'ensemble L des points M du plan tels que les triangles MAB et MAC aient la même aire.
  On note (d_0) la parallèle à (BC) passant par A et (d_1) la médiane issue de A dans ABC.
  1. Montrer que l'ensemble $(d_0) \cup (d_1)$ est inclus dans L.
     Pour tout point M distinct de A, on note d_B et d_C les distances respectives de B et C à la droite (AM).
  2. Soit M un point n'appartenant pas à (d_0). On appelle J l'intersection de la droite (AM) et de la droite (BC).
     a) Montrer que si $M \in L$ alors $d_B = d_C$.
     b) En déduire que J est le milieu de $[BC]$.
  3. Conclure sur l'ensemble L.
- Le travail demandé au candidat
  Pendant sa préparation, le candidat traitera les questions suivantes :
  Q.1) Dégager les méthodes et les savoirs mis en jeu dans la résolution de l'exercice.
  Q.2) Proposer une solution de la question 2) telle que vous la présenteriez à une classe.
- Une présentation et une solution avec `Xcas`
  1. On tape :

     ```
     A:=point(-2,-2,'couleur'=0);
     B:=point(3,1,'couleur'=0);
     C:=point(-2,3/2,'couleur'=0);
     d0:=parallele(A,droite(B,C), affichage=1);
     T:=triangle(A,B,C):;T;
     d1:=mediane(A,B,C,affichage=1);
     assume(t=[0.4,-5.0,5.0,0.0]);
     M:=element(d0,t);
     N:=element(d1,t);
     normal(aire(A,M,B));
     normal(aire(A,M,C));
     normal(aire(A,N,B);
     normal(aire(A,N,C);
     ```

     On obtient (attention avec `Xcas` les aires sont algébriques !) :



2. Soient $B1$ et $C1$ sont les projections de B et C sur la droite AM et posons : $d_B = BB1$ et $d_C = CC1$. Si les aires des triangles AMB et AMC sont égales c'est que B et C sont des points équidistant de la droite AM puisque on a :
aire(AMB)=AM*d_B=aire(AMC)=AM*d_C.
Il faut distinguer 2 cas :
   – soit B et C sont d'un même côté par rapport à AM et alors AM et BC sont paralléles car $BCC1B1$ est un rectangle.



– B et C sont de part et d'autre de AM et alors AM est la médiane du triangle ABC : en effet si AM coupe BC en J, les triangles rectangles $BB1J$ et $CC1J$ sont égaux et donc $BJ = CJ$.



On en déduit donc que $L = (d_0) \cup (d_1)$

3.17 Thème : Interprétation géométrique des nombres complexes

– L'exercice proposé au candidat

1. Les nombres complexes a_1, a_2, a_3 et a_4 sont donnés.
   Résoudre le système d'équations :
$$\begin{cases} z_1 + z_2 &= a_1 \\ z_2 + z_3 &= a_2 \\ z_3 + z_4 &= a_3 \\ z_4 + z_1 &= a_4 \end{cases}$$
   d'inconnue $(z_1, z_2, z_3, z_4) \in \mathbb{C}^4$.
2. Dans le plan, on considère un quadrilatère $A_1A_2A_3A_4$.
   Montrer qu'il existe un quadrilatère $M_1M_2M_3M_4$ dont les milieux des côtés sont les points A_1, A_2, A_3 et A_4 si et seulement si le quadrilatère $A_1A_2A_3A_4$ est un parallélogramme.
   Montrer que, dans ce cas, le point de concours des diagonales du parallélogramme $A_1A_2A_3A_4$ est l'isobarycentre des points M_1, M_2, M_3 et M_4.

– Le travail demandé au candidat
Pendant sa préparation, le candidat traitera les questions suivantes :
Q.1) Dégager les diverses étapes de la résolution de la première question de l'exercice.
Q.2) Indiquer les connaissances et savoir-faire mis en jeu dans cet exercice.

– Une présentation et une solution avec `Xcas`

1. On tape :
```
A :=syst2mat([z1+z2=a1,z2+z3=a2,z3+z4=a3,
        z4+z1=a4], [z1,z2,z3,z4])
```
   On obtient :
```
[[1,1,0,0,-a1],[0,1,1,0,-a2],[0,0,1,1,-a3],
           [1,0,0,1,-a1+a2-a3]]
```
   On tape :
```
B :=A[0..3,0..3]
```
   On obtient :
```
[[1,1,0,0],[0,1,1,0],[0,0,1,1],[1,0,0,1]]
```
   On tape :
```
det(B)
```
   On obtient :
```
0
```

ce qui veut dire qu'une ligne (par exemple la dernière ligne) est combinaison des autres lignes. On a : $z_1 + z_4 = (z_1 + z_2) - (z_2 + z_3) + (z_3 + z_4)$
Donc, il y a des solutions si on a `a4 :=a1-a2+a3`.
On tape :

```
C :=A[0..2,0..2]
```

On obtient :

```
[[1,1,0],[0,1,1],[0,0,1]]
```

On tape :

```
det(C)
```

On obtient :

```
1
```

Donc, il y a des solutions si et seulement si on a `a4 :=a1-a2+a3`. On tape :

```
a4 :=a1-a2+a3
```

```
linsolve([z1+z2=a1,z2+z3=a2,z3+z4=a3, z4+z1=a4],
[z1,z2,z3,z4])
```

On obtient :

```
[a1-a2+a3-z4,a2-a3+z4,a3-z4,z4]
```

2. On traite le problème avec les complexes Si les 4 points $M_1,\ M_2, M_3,\ M_4$ ont pour affixe z_1, z_2, z_3, z_4 Les milieux des côtés du quadrilatère $M_1M_2M_3M_4$ ont alors pour affixe b_1, b_2, b_3, b_4 on doit donc avoir :

$$\frac{z_1 + z_2}{2} = b_1, \frac{z_2 + z_3}{2} = b_2, \frac{z_3 + z_4}{2} = b_3, \frac{z_4 + z_1}{2} = b_4$$

On pose :

$$a_1 = 2b_1, a_2 = 2b_2, a_3 = 2b_3, a_4 = 2b_4$$

On a vu précédement que le système

$$\left\{ \begin{array}{ll} z_1 + z_2 & = a_1 \\ z_2 + z_3 & = a_2 \\ z_3 + z_4 & = a_3 \\ z_4 + z_1 & = a_4 \end{array} \right.$$

a des solutions si et seulement si $a_1 - a_2 + a_3 - a_4 = 0$ ou encore

$$\frac{a_1}{2} - \frac{a_2}{2} + \frac{a_3}{2} - \frac{a_4}{2} = b_1 - b_2 + b_3 - b_4 = 0$$

ou encore
$b_2 - b_1 = b_3 - b_4$ ou encore :
$\overrightarrow{A_1A_2} = \overrightarrow{A_4A_3}$ c'est à dire que le quadrilatère $A_1A_2A_3A_4$ est un parallélogramme.

3. Avec un raisonnement géométrique On ouvre un écran de géométrie exacte et on clique sur 4 points que l'on nomme `M1,M2,M3,M4` en modifiant les niveaux.
Puis, on tape :

```
A1:=milieu(M1,M2);
A2:=milieu(M3,M2);
A3:=milieu(M3,M4);
A4:=milieu(M1,M4);
polygone(M1,M2,M3,M4);
polygone(A1,A2,A3,A4,affichage=1);
```

On obtient :



On tape :

```
est_parallelogramme(A1,A2,A3,A4)
```

On obtient :

```
1
```

Réciproquement, On ouvre un écran de géométrie et on clique sur 3 points `B1,B2,B4` puis, on tape :

```
parallelogramme(B1,B2,B4,B3)
```

on clique sur un point `N1` puis, on tape :

```
N2:=symetrie(B1,N1);
N3:=symetrie(B2,N2);
N4:=symetrie(B3,N3);
N1N:=symetrie(B4,N4)
polygone_ouvert(N1,N2,N3,N4,N1N,affichage=1);
```

On obtient :



On vérifie que `N1N` et `N1` on le même affixe, on tape :

```
affixe(N1N)==affixe(N1)
```

On obtient :

```
1
```

ou on tape :

```
normal(affixe(N1N)-affixe(N1))
```

On obtient :

```
0
```